
La sémantique des langages de programmation

Quoi ?
La description mathématique du comportement des programmes.

Pourquoi ?
Mieux programmer en comprenant très précisément ses programmes.
Vérifier la correction des programmes et des outils qui les manipulent.
Guider la conception de nouveaux langages et outils.

Comment ?
Ce cours : une approche pratique, illustrée par la pratique.

Sémantique et équivalence

La question clef
Sous quelle condition peut-on dire que deux programmes sont équivalents ?

Deux notions d’égalité
Syntaxique : les programmes représentent le même arbre de syntaxe ;

Déjà pas si trivial que ça à cause des variables liées.
Sémantique : les programmes “calculent la même chose”.

Plusieurs façons de définir ce qu’un programme calcule.

Le reste de la séance : illustrer sur des langages miniatures avec du code.

Qu’est-ce qu’un programme ?

Une première approximation
L’espacement et les commentaires n’entrent pas en ligne de compte.
Un programme est donc un arbre de syntaxe abstraite.

Par exemple, les arbres de syntaxe abstraite du mini-langage L0 :
1 type t = Var of Id.t (* occurrence "x", "y", etc. *)

2 | Int of int (* entier litéral "42", etc. *)

3 | Add of t * t (* somme "e1 + e2" *)

4 | Let of t * bound1 (* déf. locale "let e1 be x in e2" *)

5 and bound1 = { bound : Id.t; body : t; } (* liaison "x.e" *)

Un problème
L’égalité brute des arbres de syntaxe abstraite est trop fine. Par exemple,

Let (Int 1, { bound = "x", body = Add (Var "x", Int 2); })

Let (Int 1, { bound = "y", body = Add (Var "y", Int 2); })

devraient être considérés équivalents. C’est la notion d’α-équivalence.

Occurrences libres et liées de variables

Définition (Occurrence d’une variable)
Soient x une variable et e une expression. Une occurrence de x dans e est
la position dans e d’une sous-expression de la forme Var x.

L’unique occurrence de x dans l’expression

Let (Int 1, { bound = "y"; body = Add (Var "x", Int 1); })

est de nature syntaxiquement différente de celle dans

Let (Int 1, { bound = "x"; body = Add (Var "x", Int 1); })

dans la mesure où, intuitivement :
la première se réfère au contexte ambiant et est dite libre, tandis que
la seconde nomme une sous-expression (Int 1) et est donc dite liée.

Seule la première peut être substituée (remplacée) si besoin est.

Calcul des variables et variables libres occurrentes

1 let rec occurring : L0.t -> Id.Set.t =

2 let open Id.Set in function

3 | Var x -> singeton x

4 | Int _ -> empty

5 | Add (e1, e2) -> union (occurring e1) (occurring e2)

6 | Let (e, b) -> union (occurring e) (occurring b.body)

1 let rec occurring_free : L0.t -> Id.Set.t =

2 let open Id.Set in function

3 | Var x -> singeton x

4 | Int _ -> empty

5 | Add (e1, e2) -> union (occurring_free e1) (occurring_free e2)

6 | Let (e, b) -> union (occurring_free e) (occurring_free_b1 b)

7 and occurring_free_b1 : L0.bound1 -> Id.Set.t =

8 fun { bound; body; } -> Id.Set.remove bound (occurring_free body)

Remarque
Id.Set.subset (occurring_free e) (occurring e) est toujours true.

L’égalité à renommage des occurrences liées près
Une occurence liée n’a pas d’identité : elle ne fait que référence à un lieur.

let x = 1 in (x + let x = 2 in x) ≡ let x = 1 in (x + let y = 2 in y)

6≡ let x = 1 in (x + let y = 2 in x)

On doit donc voir les expressions comme des graphes de syntaxe abstraite.

let • = 1 in (• + let • = 2 in •) 6= let • = 1 in (• + let • = 2 in •)

En pratique
Les implémentations qui doivent traiter sérieusement les occurrences
libres utilisent des variantes de graphes de syntaxe abstraite.
On peut se contenter des arbres de syntaxe abstraite, mais il faut alors
s’assurer de n’écrire que des opérations compatibles avec les graphes.

La substitution, visuellement

On note e[x\e′] la substitution des occurrences libres de x par e′ dans e.

e

x3x2x1

e′

y

⇒
e[x3\e′]

y

x2x1

La substitution parallèle, visuellement
On généralise pour substituer simultanément plusieurs variables.

1 type subst = (Id.t * t) list

2 let s = [(x1, e1); (x2, e2); (x3, e3)]

e

x3x2x1

x3x2
s

x1

y1 y2

⇒
e[s]

y2y1

La substitution parallèle pour L0
1 let rec subst : t -> subst -> t =

2 fun e s ->

3 match e with

4 | Var x -> (try List.assoc x s with Not_found -> e)

5 | Int _ -> e

6 | Add (e1, e2) -> Add (subst e1 s, subst e2 s)

7 | Let (e, b) -> Let (subst e s, subst_b1 b s)

8 and subst_b1 : bound1 -> subst -> bound1 =

9 fun { bound = x; body; } s ->

10 let x' = Var (Id.fresh ()) in

11 { bound = x'; body = subst body ((x, x') :: s); }

Deux écueils évités par le renommage systématique des variables liées
La substitution erronnée d’une occurrence liée dans e.

(List.assoc assure un fonctionnement last in, first out.)
La capture les variables libres de s en passant sous un lieur de e.

L’α-équivalence pour L0

1 let close_with : bound1 -> t -> t =

2 fun { bound; body; } e -> subst body [(bound, e)]

3

4 let rec equal e e' =

5 match e, e' with

6 | Var x, Var x' -> x = x'

7 | Int i, Int i' -> i = i'

8 | Add (e1, e2), Add (e1', e2') -> equal e1 e1' && equal e2 e2'

9 | Let (e, b), Let (e', b') -> equal e e' && equal_b1 b b'

10 | _ -> false

11

12 and equal_b1 b b' =

13 let y = Var (Id.fresh ()) in

14 equal (close_with b y) (close_with b' y)

15 / 50



Plan

1 Introduction

2 Équivalence syntaxique et substitution

3 Évaluation
Évaluation naïve
Environnements et fermetures
Métacircularité
Évaluation des références
Évaluation du filtrage de motifs

4 Conclusion



Plan

1 Introduction

2 Équivalence syntaxique et substitution

3 Évaluation
Évaluation naïve
Environnements et fermetures
Métacircularité
Évaluation des références
Évaluation du filtrage de motifs

4 Conclusion



Vers des notions d’équivalence sémantiques

Une première intuition
Deux programmes sont équivalents quand “ils ont le même résultat”.
Il faut donc définir le résultat d’un programme, i.e., de son évaluation.

Il y a de nombreuses façons de définir ce résultat :
dans un style à grands pas, par une fonction partielle d’évaluation ;
dans un style à petit pas, par une relation de réduction ;
…

Chacune de ces approches admet de nombreuses variations.

On va voir les deux approches en OCaml, pour aboutir à un interprète.

18 / 50

Un évaluateur pour L0
Il est facile d’évaluer les expressions closes (sans variables libres) de L0.

1 type value = int

2

3 let rec eval : t -> value = function

4 | Var _ -> invalid_arg "open expression"

5 | Int n -> n

6 | Add (e1, e2) -> eval e1 + eval e2

7 | Let (e, b) -> eval (close_with b e)

Voyez-vous une alternative pour l’évaluation des définitions locales ?
1 let rec eval : t -> value = function

2 | Var _ -> invalid_arg "open expression"

3 | Int n -> n

4 | Add (e1, e2) -> eval e1 + eval e2

5 | Let (e, b) -> eval (close_with b (Int (eval e)))

Le second choix est majoritaire dans les langages de programmation.

Le langage L1

On étend la syntaxe de L0 avec des fonctions de première classe.
1 type t = Var of Id.t (* occurrence "x", "y", etc. *)

2 | Int of int (* entier litéral "42", etc. *)

3 | Add of t * t (* somme "e1 + e2" *)

4 | Let of t * bound1 (* déf. locale "let e1 be x in e2" *)

5 | Fun of bound1 (* fonction anonyme "fun x -> e" *)

6 | App of t * t (* application "e1 e2" *)

7 and bound1 = { bound : Id.t; body : bound1; }

La substitution et l’α-conversion s’étendent de façon routinière.

Un évaluateur pour L1
1 type value = VInt of int | VFun of bound1

2

3 let rec eval : t -> value = function

4 | Var _ -> invalid_arg "open expression"

5 | Int n -> VInt n

6 | Add (e1, e2) -> add_values (eval e1) (eval e2)

7 | Let (e, b) -> eval (close_with b (reify (eval e)))

8 | Fun b -> VFun b

9 | App (e1, e2) -> app_values (eval e1) (eval e2)

10

11 and add_values v1 v2 = match v1, v2 with

12 | VInt n1, VInt n2 -> VInt (n1 + n2)

13 | _ -> failwith "ill-typed"

14

15 and app_values v1 v2 = match v1 with

16 | VFun b -> eval (close_with b (reify v2))

17 | _ -> failwith "ill-typed"

18

19 and reify : value -> expr = function VInt n -> Int n | VFun b -> Fun b

Exercice : ajouter les booléens à L1
1 type t = Var of Id.t (* occurrence "x", "y", etc. *)

2 | Int of int (* entier litéral "42", etc. *)

3 | Add of t * t (* somme "e1 + e2" *)

4 ...

5 | Bool of bool (* booléen litéral "true", "false" *)

6 | If of t * t * t (* conditionnelle *)

1 type value = VInt of int | VFun of bound1 | VBool of bool

1 let rec eval : t -> value = function

2 | Var _ -> invalid_arg "open expression"

3 | Int n -> VInt n

4 | Add (e1, e2) -> add_values (eval e1) (eval e2)

5 ...

6 | Bool b -> VBool b

7 | If (e1, e2, e3) -> if_values (eval e1) e2 e3

8 and if_values v e2 e3 = match v with

9 | VBool true -> eval e1 | VBool false -> eval e2

10 | _ -> failwith "ill-typed"

Interlude

Remarque de fond : OCaml 6= Mathématiques
Écrire l’évaluateur en OCaml est agréable, le code est très concis.
En contrepartie, comprendre la sémantique de L0 telle que définie par
l’évaluateur exige de comprendre la sémantique d’OCaml.
Un cours de sémantique utilisera plutôt le langage mathématique.

Différent d’OCaml, p. ex., pas de fonctions récursives arbitraires !

1 let b : bool = let rec f () = not (f ()) in f ()

Est-ce que b est le booléen true ou le booléen false ? Est-ce que les
expressions OCaml de type bool sont en bijection avec B = {true, false} ?

Pour les curieux et les curieuses
Sémantique des langages de programmation au S2 traite de ces questions.

Des substitutions aux environnements

1 let close_with : bound1 -> t -> t =

2 fun { bound; body; } e -> subst body [(bound, e)]

3 ...

4 and app_values v1 v2 = match v1 with

5 | VFun b -> eval (close_with b (reify v2))

6 | _ -> failwith "ill-typed"

L’appel à subst dans close_with est coûteux : on parcourt tout b.body,
même les sous-expressions dont l’évaluation ne sera pas nécessaire.

Une alternative à la substitution : les environnements
L’évaluateur reçoit, en plus de l’expression, un environnement.
C’est une “substitution de valeurs”, i.e., (Id.t * value) list.
Il détermine la valeur des variables libres de l’expression évaluée.
On l’étend lors du passage sous un lieur (let, fun).

Un évaluateur avec environnement pour L1

(FAUX)

1 type env = (Id.t * value) list

2 type value = VInt of int | VFun of bound1 | VBool of bool

3 let rec eval' : env -> t -> value = fun env e -> match e with

4 | Var x -> List.assoc x env

5 | Int n -> VInt n

6 | Add (e1, e2) -> add_values' (eval' env e1) (eval' env e2)

7 | Let (e, b) -> eval'_bound1 env b (eval' env e)

8 | Fun b -> VFun b

9 | App (e1, e2) -> app_values' env (eval' env e1) (eval' env e2)

10 | Bool b -> VBool b

11 | If (e1, e2, e3) -> if_values' env (eval' env e1) e2 e3

12

13 ...

14

15 and app_values' env v1 v2 = match v1 with

16 | VFun b -> eval'_bound1 env b v2

17 | _ -> failwith "ill-typed"

18

19 and eval'_bound1 env { bound; body; } v =

20 eval' ((bound, v) :: env) body

Un évaluateur avec environnement

Pourquoi la fonction eval' est-elle fausse ? Elle doit passer ce test :
1 let test env e =

2 let subst_of_env = List.map (fun (x, v) -> (x, reify v)) in

3 eval' env e = eval (subst (subst_of_env env) e)

Que se passe-t-il si on évalue les expressions ci-dessous ?
1 let f = let x = 1 in fun y -> x + y in f 1;;

2 let g = let x = 1 in fun y -> x + y in let x = 2 in g 1;;

Environnements et portée lexicale
La gestion de l’environnement est fausse pour les fonctions.
L’environnement pertinent est celui du point de création.
Il faut transmettre cet environnement jusqu’au point d’application.

Un évaluateur avec environnement et fermetures
1 type env = (Id.t * value) list

2 and value = VInt of int | VFun of bound1 * env | VBool of bool

3

4 let rec eval' : env -> t -> value = fun env e -> match e with

5 | Var x -> List.assoc x env

6 | Int n -> VInt n

7 | Add (e1, e2) -> add_values' (eval' env e1) (eval' env e2)

8 | Let (e, b) -> eval'_bound1 env b (eval' env e)

9 | Fun b -> VFun (b, env)

10 | App (e1, e2) -> app_values' (eval' env e1) (eval' env e2)

11 | Bool b -> VBool b

12 | If (e1, e2, e3) -> if_values' env (eval' env e1) e2 e3

13 ...

14 and app_values' v1 v2 = match v1 with

15 | VFun (b, env) -> eval'_bound1 env b v2

16 | _ -> failwith "ill-typed"

17

18 and eval'_bound1 env { bound; body; } v =

19 eval' ((bound, v) :: env) body

Langage objet et langage sujet (“métalangage”)

La semaine dernière
Un évaluateur avec environnements gérant correctement la portée lexicale
via les fermetures, qui groupent les expressions et leurs environnements.

Y avait-il plus simple ?

Une solution fainéante
On utilise les booléens d’OCaml pour interpréter les booléens de L1.
On utilise les entiers d’OCaml pour interpréter les entiers de L1.
Pourquoi ne pas faire de même pour les fonctions ?

Un évaluateur avec env., partiellement métacirculaire
1 type value = VInt of int | VFun of (value -> value) | VBool of bool

2 type env = (Id.t * value) list

1 let rec eval' : env -> t -> value = fun env e -> match e with

2 | Var x -> List.assoc x env

3 | Int n -> VInt n

4 | Add (e1, e2) -> add_values' (eval' env e1) (eval' env e2)

5 | Let (e, b) -> eval'_bound1 env b (eval' env e)

6 | Fun b -> VFun (fun v_arg -> eval_bound1 env b v_arg)

7 | App (e1, e2) -> app_values' (eval' env e1) (eval' env e2)

8 | Bool b -> VBool b

9 | If (e1, e2, e3) -> if_values' env (eval' env e1) e2 e3

10 ...

11 and app_values' v1 v2 = match v1 with

12 | VFun f -> f v2

13 | _ -> failwith "ill-typed"

14

15 and eval'_bound1 env { bound; body; } v =

16 eval' ((bound, v) :: env) body

Les évaluateurs métacirculaires

Soient Ls et Lo deux langages de programmation.

Définition (Évaluateur métacirculaire)
Un évaluateur métacirculaire de Lo en Ls interprète chaque
construction de Lo par la même construction dans Ls .
Dans le cas extrême, Ls = Lo et on parle d’autoévaluateur.

De façon générale, il existe différents degrés de métacircularité.
Un interprète peut reposer plus ou moins fortement sur le métalangage.

Par exemple, nous aurions pu éviter complètement la substitution.
Solution sans environnements, ni fermetures, ni… variables !

Syntaxe abstraite d’ordre supérieur et métacircularité
1 type t = ... (* plus de constructeur Var ! *)

2 | Fun of bound1 (* fonction anonyme "fun x -> e" *)

3 | App of t * t (* application "e1 e2" *)

4 and bound1 = t -> t (* plus de substitution ! *)

5

6 type value = VInt of int | VFun of (value -> value) | VBool of bool

1 let rec eval : t -> value = function

2 | Int n -> VInt n

3 | Add (e1, e2) -> add_values (eval e1) (eval e2)

4 | Let (e, b) -> eval_bound1 b (eval e)

5 | Fun b -> VFun (eval_bound1 b)

6 | App (e1, e2) -> app_values (eval e1) (eval e2)

7 ...

1 and eval_bound1 bound1 v = eval (bound1 (reify v))

2

3 and reify = function VInt n -> Int n | VBool -> Bool b

4 | VFun f -> Fun (fun x -> reify (f (eval x)))

Bilan au sujet des évaluateur métacirculaires
Un évaluateur métacirculaire :

permet de réaliser des extensions de langage à faible coût (cf. Lisp) ;
Interprétation métacirculaire de Lo = Ls ∪ {nouvelles constructions}.

délègue une bonne partie de la sémantique de Lo à celle de Ls .
Y compris les aspects subtils et implicites, e.g., l’ordre d’évaluation.

Observation générale
Plus Ls et Lo sont éloignés, plus l’interprète est informatif.

Exemple
L’évaluateur avec environnements/fermetures, qui fonctionnerait dans un
langage de premier ordre, explique la portée bien plus que le précédent.

Retour sur une remarque précédente
Cas intéressant : Ls est le langage mathématique, expressif et restreint.

Le langage L2
On ajoute à L1 :

des références qui peuvent contenir des valeurs arbitraires ;
la séquence avec une valeur triviale pour élément neutre (type unit).

1 type t = Var of Id.t (* occurrence "x", "y", etc. *)

2 | Int of int (* entier litéral "42", etc. *)

3 | Add of t * t (* somme "e1 + e2" *)

4 | Bool of bool (* booléen litéral "true", "false" *)

5 | If of t * t * t (* conditionnelle *)

6 | Let of t * bound1 (* déf. locale "let e1 be x in e2" *)

7 | Fun of bound1 (* fonction anonyme "fun x -> e" *)

8 | App of t * t (* application "e1 e2" *)

9 | Ref of t (* allocation "ref e" *)

10 | Read of t (* déréférencement "!e" *)

11 | Asgn of t * t (* assignation "e1 := e2" *)

12 | Unit (* 0-uplet *)

13 | Seq of t * t (* séquence "e1; e2" *)

14 and bound1 = { bound : Id.t; body : bound1; }

36 / 50



Un évaluateur métacirculaire pour L2
1 type env = (Id.t * value) list

2 and value = VInt of int | VBool of bool | VUnit

3 | VFun of bound1 * env | VRef of value ref

4

5 let rec eval : env -> t -> value = fun env e -> match e with

6 | Add (e1, e2) -> add_values (eval env e1) (eval env e2)

7 | Ref e -> VRef (ref (eval env e))

8 | Read e -> read_value (eval env e)

9 | Asgn (e1, e2) -> asgn_values (eval env e1) (eval env e2)

10 | Unit -> VUnit

11 | Seq (e1, e2) -> seq_value (eval env e1) env e2

12 ...

13

14 and read_value = function VRef r -> !r | _ -> failwith "ill-typed"

15

16 and asgn_values v1 v2 =

17 match v1 with VRef r -> r := v2; VUnit | _ -> failwith "ill-typed"

18

19 and seq_value v1 env e2 =

20 match v1 with VUnit -> eval env e2 | _ -> failwith "ill-typed"

Références et métacircularité
1 let e =

2 let x = Var "x" in

3 Let (Ref (Int 0), { bound = "x";

4 body = Add (Seq (Asgn (x, Int 1), Read x), Read x); })

Vers quel entier s’évalue l’expression e ?

utop # open Lang2;;

utop # let e = let x = Var "x" in Let (Ref (Int 0), ...);;

val e : t = let(ref 0, x.(((x := 1); !x) + !x))

utop # eval [] e;;

- : value = 1

Qu’est-ce qui explique ce résultat dans la définition de eval ?

Il dépend de l’ordre d’évaluation des arguments de fonction en OCaml.
Formellement, celui-ci n’est pas défini. Donc celui de L2 non plus !

Cf. section Basic expressions (7.2) du manuel d’OCaml.

En pratique, OCaml évalue les arguments de droite à gauche.

Cf. le rapport de recherche The ZINC Experiment de X. Leroy.

Formellement, celui-ci n’est pas défini. Donc celui de L2 non plus !

Cf. section Basic expressions (7.2) du manuel d’OCaml.

En pratique, OCaml évalue les arguments de droite à gauche.

Cf. le rapport de recherche The ZINC Experiment de X. Leroy.

Et en OCaml ?
L’évaluateur métacirculaire n’explique aucunement les références.
Pour les comprendre, il est utile de réfléchir à la sémantique d’OCaml.

Donner les valeurs des expressions suivantes, placées après let x = ref 0.

let x = 1 in (let x = 2 in x) + x ≡

3

x := 1; (x := 2; !x) + !x ≡

3

x := 1; !x + (x := 2; !x) ≡

4

let y = x in y := 1; !x + !y ≡

2

Observations immédiates
L’assignation n’est pas locale ; elle ne repose pas sur la substitution.
L’évaluation des deux dernières expressions n’est pas une pure valeur.

Elle modifie le contenu de x pour la suite du programme.
Ces expressions “impures” sont sensibles à l’ordre d’évaluation.

Gestion de la mémoire
Les références manipulent des adresses abstraites dans une mémoire.

1 module Address : sig

2 type t (* un type de données abstrait... *)

3 val compare : t -> t -> int (* ... inextinguible ... *)

4 val fresh : unit -> t (* ... et totalement ordonné. *)

5 end

1 module Memory = struct

2 include Map.Make(Address)

3 let alloc v0 m = let a = Address.fresh () in a, add a v0 m

4 let read a m = find a m

5 let asgn a v m = if mem a m then add a v mem else raise Not_found

6 end

Remarque : un modèle abstrait de la mémoire
Ici la mémoire n’est pas un tableau, ni les adresses des indices.
En particulier, l’arithmétique de pointeur n’est pas possible.

Évaluation des références (1/2)

1 type env = (Id.t * value) list and value = VInt of int

2 | VBool of bool

3 | VFun of bound1 * env

4 | VRef of Address.t

5

6 type mem = value Memory.t

Quel serait le type d’un évaluateur écrit en OCaml purement fonctionnel ?

1 val eval : env -> t -> mem -> value * mem

N.B. : ne suppose pas l’existence de références dans le métalangage !

41 / 50



Évaluation des références (1/2)

1 type env = (Id.t * value) list and value = VInt of int

2 | VBool of bool

3 | VFun of bound1 * env

4 | VRef of Address.t

5

6 type mem = value Memory.t

Quel serait le type d’un évaluateur écrit en OCaml purement fonctionnel ?

1 val eval : env -> t -> mem -> value * mem

N.B. : ne suppose pas l’existence de références dans le métalangage !

41 / 50



Évaluation des références (1/2)

1 type env = (Id.t * value) list and value = VInt of int

2 | VBool of bool

3 | VFun of bound1 * env

4 | VRef of Address.t

5

6 type mem = value Memory.t

Quel serait le type d’un évaluateur écrit en OCaml purement fonctionnel ?

1 val eval : env -> t -> mem -> value * mem

N.B. : ne suppose pas l’existence de références dans le métalangage !

41 / 50



Évaluation des références (2/2)
1 let rec eval env e m =

2 match e with

3 | Var x -> List.assoc x env, m

4 | Add (e1, e2) -> let v1, m = eval env e1 m in

5 let v2, m = eval env e2 m in

6 add_values v1 v2, m

7 | Fun b -> VFun (b, env), m

8 ...

9 | Ref e -> let v, m = eval env e m in

10 let a, m = Memory.alloc v m in VRef a, m

11 | Read e -> let v, m = eval env e m in read_value v m

12 | Asgn (e1, e2) -> let v1, m = eval env e1 m in

13 let v2, m = eval env e2 m in

14 asgn_values v1 v2 m

15

16 and read_value v m = match v with

17 | VRef a -> Memory.read a m, m | _ -> failwith "ill-typed"

18

19 and asgn_values v1 v2 m = match v1 with

20 | VRef a -> VUnit, Memory.asgn a v2 m | _ -> failwith "ill-typed"

Programmer mieux pour programmer moins
1 let rec eval : env -> t -> mem -> value * mem = fun env e m ->

2 match e with

3 | Var x -> List.assoc x env, m

4 | Add (e1, e2) -> let v1, m = eval env e1 m in

5 let v2, m = eval env e2 m in

6 add_values v1 v2, m

7 | Fun b -> VFun (b, env), m

8 ...

De quelques effets désagréables du changement de type de eval

Nous avons dû modifier l’interprétation de chaque construction.
Les cas préexistants sont pollués par de la bureaucratie.

Opérations logistiques récurrentes
1 Renvoyer une valeur et la mémoire non-modifiée.

2 Chaîner les transformations de la mémoire.

L’abstraction monadique

1 type 'a stateful = mem -> 'a * mem

2 val return : 'a -> 'a stateful

3 val ( >>= ) : 'a stateful -> ('a -> 'b stateful) -> 'b stateful

1 let rec eval : env -> t -> value stateful = fun env e ->

2 match e with

3 | Var x -> return (List.assoc x env)

4 | Add (e1, e2) -> eval env e1 >>= fun v1 ->

5 eval env e2 >>= fun v2 ->

6 return (add_values v1 v2)

7 | Fun b -> return (VFun (b, env))

8 ...

1 let return x = fun m -> x, m

2 let ( >>= ) xm f = fun m -> let x, m = xm m in f x m

Les opérateurs de liaison définis par l’utilisateur
En OCaml 4.08+ (2019)
La notation “let+ x = e1 in e2” désigne “(let+) e1 (fun x -> e2)”.

1 let ( let+ ) = ( >>= )

2

3 let rec eval : env -> t -> value stateful = fun env e ->

4 match e with

5 | Var x -> return (List.assoc x env)

6 | Add (e1, e2) -> let+ v1 = eval env e1 in

7 let+ v2 = eval env e2 in

8 return (add_values v1 v2)

9 | Fun b -> return (VFun (b, env))

10 ...

Bilan
Le code est très lisible et proche du code original.
L’opérateur (let+)/(>>=) nous force à expliciter l’ordre d’évaluation.

Le langage L3

On ajoute à L1 les n-uplets et le filtrage de motifs.

1 type t = Var of Id.t (* occurrence "x", "y", etc. *)

2 | Int of int (* entier litéral "42", etc. *)

3 | Add of t * t (* somme "e1 + e2" *)

4 ...

5 | Tuple of t list (* n-uplets "(e1, ..., eN)" *)

6 | Match of exp * branch list (* filtrage "match e with ..." *)

1 and branch = pattern * exp (* branche "| p -> e" *)

1 and pattern = PVar of Id.t (* motif universel liant *)

2 | PWildcard (* motif universel non liant *)

3 | PInt of int (* motif constant (entier) *)

4 | PBool of bool (* motif constant (booléen) *)

5 | PTuple of pattern list (* motif n-uplet *)

Un évaluateur pour L3
1 type env = (Id.t * value) list

2 and value = VInt of int | VBool of bool

3 | VFun of bound1 * env | VTuple of value list

4

5 let rec eval : env -> t -> value = fun env e -> match e with

6 | Var x -> List.assoc x env

7 | Int n -> VInt n

8 | Add (e1, e2) -> add_values (eval env e1) (eval env e2)

9 ...

10 | Tuple es -> VTuple (List.map (eval env) es)

11 | Match (e, bs) -> match_value (eval env e) env bs

12

13 and match_value v env bs = (* à faire pour le jalon 2 ! *)

La sémantique de match v with p1 -> e1 | ... | pN -> eN

On évalue la première expression eI telle que pI accepte la valeur v.
L’évaluation procède dans l’environnement étendu par pI.

val filter_value : pattern -> value -> env option

10 | Tuple es -> VTuple (List.map (eval env) es)

11 | Match (e, bs) -> match_value (eval env e) env bs

12

13 and match_value v env bs = (* à faire pour le jalon 2 ! *)

La sémantique de match v with p1 -> e1 | ... | pN -> eN

On évalue la première expression eI telle que pI accepte la valeur v.
L’évaluation procède dans l’environnement étendu par pI.

val filter_value : pattern -> value -> env option

2 Équivalence syntaxique et substitution

3 Évaluation

Environnements et fermetures
Métacircularité
Évaluation des références

Conclusion au sujet de l’interprétation

Les évaluateurs définitionnels
Évaluer une expression par parcours ascendant de sa syntaxe.
Inefficaces, mais simples : une définition exécutable.

Attention à ne pas trop reposer sur le métalangage !
Il existe des techniques d’interprétation beaucoup plus efficaces.

Souvent via une machine abstraite, cf. ocamlrun, CPython, Marthe…

Le jalon 2 : un interprète définitionnel pour Hopix
Des types construits et des enregistrements.
Les détails du filtrage de motifs.
Les fonctions (mutuellement) récursives.

La prochaine fois : la vérification de types.

